


RING – Example reaction rules

1. Hydrogenation of C=C




Code:

```
rule Hydrogenation{
  reactant r1 {
    C labeled c1
    C labeled c2 double bond to c1
  }
  reactant r2 {
    H labeled h1
    H labeled h2 single bond to h1
  }
  break bond (h1, h2)
  decrease bond order (c1, c2)
  form bond (c1, h1)
  form bond (c2, h2)
}
```

Explanation:

1. Define two reactants – carbon-carbon double bonds and Hydrogen.
2. No explicit constraints mentioned here
3. Transformations include (a) breaking Hydrogen-Hydrogen single bond, changing carbon-carbon double bond into single bond and forming single bonds between the two C-H pairs.

2. Hydration



Code:

```
rule Hydration{
  positive reactant r1 {
    C+ labeled c1
```

```

    C labeled c2 single bond to c1
  }
  reactant r2 {
    O labeled o1
  }
  form bond(c1, o1)
  modify atomtype (c1, C)
  modify atomtype (o1, O+)
}

```

Explanation:

1. define two reactants – a positive carbenium ion and a neutral oxygen.
2. No constraints specified
3. Oxygen forms oxonium, while the carbenium loses its charge upon the formation of the C-O bond.

Note that, per the definition, the oxygen could be a water molecule, alcohol, or ethers.

3. Homolytic Scission



Code:

```


rule Homocyclicscission{
  reactant r1 {
    C labeled c1
    C labeled c2 single bond to c1
  }
  break bond(c1, c2)
  modify atomtype (c1, C.)
  modify atomtype (c2, C.)
}

```

Explanation:

The carbon-carbon single bond cleaves homolytically to form two radicals.

4. Esterification




Code:

```
rule Esterification{
  reactant r1{
    C labeled c1
    O labeled o1 single bond to c1
    H labeled h1 single bond to o1
    O labeled o2 double bond to c1
  }
  reactant r2{
    C labeled c2
    O labeled o3 single bond to c2
    H labeled h2 single bond to o3
  }
  break bond(c1, o1)
  break bond(o3, h2)
  form bond(c1, o3)
  form bond(o1, h2)
}
```

Explanation:

1. r1 is the acid, while r2 is the alcohol
2. The reaction proceeds with acid hydroxy group lost as water per the reaction rule defined because the acid C-O single bond breaks, while a single bond between carbonyl C and alcohol O is formed. A water molecule is lost in the process.

5. Etherification



Code:

```
rule Etherification{
```

```


positive reactant r1{
  C+ labeled c1
  C labeled c2 single bond to c1
}
reactant r2{
  C labeled c3
  O labeled o1 single bond to c3
  H labeled h1 single bond to o1
}
break bond(o1, h1)
form bond(c1, o1)
modify atomtype(c1, C)
modify atomtype(h1, H+)
}

```

Explanation:

1. r1 is a carbenium ion attached to another carbon with a single bond while r2 is an alcohol
2. No constraints here as well
3. Transformations include (a) forming the C-O bond, carbenium ion losing its charge, and hydroxy hydrogen leaving as a proton

6. Aldol Condensation



```

rule Aldolcondensation{
  reactant r1{
    C labeled c1
    C labeled c2 single bond to c1
    C labeled c3 single bond to c2

```

```

    O labeled o1 double bond to c1
    H labeled h1 single bond to c2
    H labeled h2 single bond to c2
  }
reactant r2{
    C labeled c4
    O labeled o2 double bond to c4
  }


break double bond (c4, o2)
break bond (c2, h1)
break bond (c2, h2)
form bond (o2, h1)
form bond (o2, h2)
form double bond (c2, c4)
}

```

Explanation:

1. r1 is an carbonyl-containing molecule with the active methylene group, while r2 is the second carbonyl-containing molecule that loses its oxygen.
2. The oxygen of the second carbonyl group is lost as water along with the hydrogens from the first reactant.
3. This rule breaks the double-bond of a C=O group directly as well as forms a C=C directly.

7. Oligomerization of olefins



Code:

```

rule Oligomerization{
  reactant r1{

```

```

    C labeled c1
    C labeled c2 double bond to c1
  }
  reactant r2{
    C labeled c3
    C labeled c4 double bond to c3
    H labeled h1 single bond to c3
  }
  decrease bond order (c1, c2)
  form bond(c2, c3)
  break bond (h1,c3)
  form bond (h1,c1)
  allow intramolecular reactions
  product constraints on mol{
    (mol is cyclic && mol.minringsize >=5 ) || (! mol is cyclic)
  }


```

Explanation:

1. Reactants are two molecules with C-C double bonds
2. C-C bond is formed between r1 and r2.
3. The hydrogen from the second reactant moves to the first reactant
4. Intramolecular reactions are also possible – RING will check for the possibility of an intramolecular reaction.
5. A product constraint is added so that no products formed have a ring of size < 4 atoms!

8. Hydrogenation of acid with two H₂

Constraints: No C=C or $\text{C}=\text{C}$ or $\text{C}-\overset{\text{O}}{\parallel}{\text{C}}-\text{C}$ or $\text{C}-\overset{\text{O}}{\parallel}{\text{C}}-\text{H}$ in the reactant.



Code:

```

rule HydrogenationAcidToAlcohol{
  reactant r1{
    C labeled c1
    O labeled o1 double bond to c1
    O labeled o2 single bond to c1
    H labeled h1 single bond to o2
  }
  reactant r2{
    H labeled h2
    H labeled h3 single bond to h2
  }
  reactant r3 duplicates r2 (h2=>h4, h3=>h5)
  constraints{
    fragment e{
      C labeled c2
      C labeled c3 double bond to c2
    }
    fragment f{
      C labeled c4{! connected to O with single bond}
      O labeled o3 double bond to c4
    }
    ! r1 contains e
    ! r1 contains f
  }
  break bond (c1, o2)

```

decrease bond order (c1, o1)

break bond (h2, h3)

break bond (h4, h5)

form bond (h2, o1)

form bond (h3, o2)

form bond (h4, c1)


form bond (h5, c1)

}

Explanation:

1. First reactant is an acid, while the second is hydrogen molecule. The third reactant is another hydrogen molecule which is described as a duplicate of the second reactant. Note the label mapping '=>'.
2. The constraints are structural in nature – no C=C, and no carbonyl group belonging to keto or aldehyde should be present in the molecule
3. 'e' described the C=C fragment, while 'f' describes a carbonyl group such that the carbon is not attached to any other oxygens.

9. Oxidation of Aldehyde



Code:

```
rule OxidationofAldehyde{  
  reactant r1{  
    C labeled c1  
    C labeled c2 single bond to c1  
    O labeled o1 double bond to c1  
    H labeled h1 single bond to c1  
  }  
  reactant r2{
```



```


O labeled o2
O labeled o3 double bond to o2
}
reactant r3 duplicates r1 (c1=>c3, c2=>c4, o1=>o4, h1=>h2)
break bond (h1, c1)
break bond (h2, c3)
break double bond (o2, o3)
form bond (c1, o2)
form bond (o2, h1)
form bond (c3, o3)
form bond (o3, h2)
}

```

Explanation:

1. The stoichiometric reaction requires two moles of aldehyde per mole of oxygen molecule.
2. The first reactant is an aldehyde, the second is oxygen molecule, while the third reactant duplicates the first reactant
3. The transformations involve breaking the two oxygens of the oxygen molecule, and formation of the C-O bond to ultimately form the acid.

10. Sulfonation



Code:

```

rule sulfonation{
  neutral reactant arom{
    c labeled c1
    c labeled c2 aromatic bond to c1

```

```

    }
    positive reactant sulf{
        S labeled s1 {connected to 2 O with double bond}
        O+ labeled o1 double bond to s1
    }
    form bond (c1,s1)
    decrease bond order(s1,o1)
    modify bond (c1,c2,single)
    modify atomtype(c2,C+)
    modify atomtype (c1,C)
    modify atomtype (o1, O)
}

```

Explanation:

1. Reactant r1 is a molecule with a c-c aromatic bond, while r2 is the electrophile – protonated sulfur trioxide.
2. Transformations involve adding the positive charge to one carbon, modifying the atomtypes and bond to non aromatic (C+, C, single bond, respectively), while the oxonium loses its charge and the S=O+ bond becomes single bonded.